

ルービックキューブと Gap

福岡教育大学 情報教育教室

藤本 光史*

平成18年7月10日

1 ルービックキューブと置換

ルービックキューブの各面に以下のように番号を付ける。

| | | | | | | | | | | | |
|----|-------|----|----|--------|----|----|------|----|----|--------|----|
| | | | 1 | 2 | 3 | | | | | | |
| | | | 4 | Red | 5 | | | | | | |
| | | | 6 | 7 | 8 | | | | | | |
| 9 | 10 | 11 | 17 | 18 | 19 | 25 | 26 | 27 | 33 | 34 | 35 |
| 12 | White | 13 | 20 | Green | 21 | 28 | Blue | 29 | 36 | Yellow | 37 |
| 14 | 15 | 16 | 22 | 23 | 24 | 30 | 31 | 32 | 38 | 39 | 40 |
| | | | 41 | 42 | 43 | | | | | | |
| | | | 44 | Orange | 45 | | | | | | |
| | | | 46 | 47 | 48 | | | | | | |

これによって、ルービックキューブの様々なパターンが置換を用いて表現できる。

*fujimoto@fukuoka-edu.ac.jp, <http://www.fukuoka-edu.ac.jp/~fujimoto/>

2 ルービックキューブ群の定義

群論用数式処理ソフト Gap¹ に群を定義する。まず、各面を時計と逆周りに 90 度回転させる操作を次のように記述する。

```
g := (17, 19, 24, 22)(18, 21, 23, 20)(6, 25, 43, 16)
(7, 28, 42, 13)(8, 30, 41, 11);
y := (33, 35, 40, 38)(34, 37, 39, 36)(3, 9, 46, 32)
(2, 12, 47, 29)(1, 14, 48, 27);
b := (25, 27, 32, 30)(26, 29, 31, 28)(3, 38, 43, 19)
(5, 36, 45, 21)(8, 33, 48, 24);
w := (9, 11, 16, 14)(10, 13, 15, 12)(1, 17, 41, 40)
(4, 20, 44, 37)(6, 22, 46, 35);
r := (1, 3, 8, 6)(2, 5, 7, 4)(9, 33, 25, 17)
(10, 34, 26, 18)(11, 35, 27, 19);
o := (41, 43, 48, 46)(42, 45, 47, 44)(14, 22, 30, 38)
(15, 23, 31, 39)(16, 24, 32, 40);
```

これらを生成元とする置換群を Gap 上で定義するには、次のようにすればよい。


```
gap> cube := Group(g,y,b,w,r,o);
```

ちなみに、この群の位数は次で求められる。

```
gap> Size(cube);
```

3 変形の可能性の判定

変形したいパターンの置換表現から、それが、ルービックキューブ群に含まれるかチェックする。もし、含まれれば変形可能、そうでなければ変形不可である。



```
e.g.) gap> (8,19,25)(24,43,30) in cube;
false
gap> (8,19,25)(24,30,43) in cube;
true
```

※ (8,19,25) は 8 の場所に 19 が、19 の場所に 25 が、25 の場所に 8 が来ることを表している。

¹<http://www-gap.dcs.st-and.ac.uk/~gap/>

4 生成元の積による表示

変形可能なパターンについて、それを得るための操作を生成元 g, y, b, w, r, o の積で表現する。そのために、次の関数 `GetWordOfElements` を Gap で定義して利用する。

```
GetWordOfElements:=function(G,GenName,x)
  local gen,F,hom;
  F:=FreeGroup(GenName);
  gen:=GeneratorsOfGroup(G);
  hom:=GroupHomomorphismByImages(F,G,GeneratorsOfGroup(F),gen);
  return PreImagesRepresentative(hom,x);
end;
```

$(8,19,25)(24,30,43)$ で表されるパターンは、前節で計算したように、変形可能である。これを生成元の積で表すには、次のようにする。

```
gap> p:=GetWordOfElements(cube,["g","y","b","w","r","o"],
(8,19,25)(24,30,43));
g*r*w*g*w^-1*g^-1*r^-1*g^-1*b^-1*g^-1*o^-1*g*o*b*g^2*b*g*b^-1
*r*g^-1*r^-1*g*b*g^-1*b^-1*g*r^-1*b*r*b^-1*g
```

5 元に戻すには

前節で得られたものの逆元を求めると、それは、元に戻すための操作を表す。実際にルービックキューブを操作してみて確認しよう。できれば、その操作を入力するだけで、自動的にアニメーションで変形してくれるソフトがあるとよいが、これについてはまだ作成できていない。

```
gap> p^-1;
g^-1*b*r^-1*b^-1*r*g^-1*b*g*b^-1*g^-1*r*g*r^-1*b*g^-1*b^-1
*g^-2*b^-1*o^-1*g^-1*o*g*b*g*r*g*w*g^-1*w^-1*r^-1*g^-1
```

※注意：実際の操作は右から順に実行していく。

演習問題

1. $(13,28,29)(20,21,36)$ で表現されるパターンを図示せよ。また、このパターンが変形可能かどうか判定し、変形可能であれば、元に戻す操作を記述せよ。
2. $(13,21,29,26,45)(20,28,36,5,31)$ で表現されるパターンを図示せよ。また、このパターンが変形可能かどうか判定し、変形可能であれば、元に戻す操作を記述せよ。
3. 変形不可能なパターンを1つ作れ。

| | | | | | | | | | | | |
|----|-------|----|----|--------|----|----|------|----|----|--------|----|
| | | | 1 | 2 | 3 | | | | | | |
| | | | 4 | Red | 5 | | | | | | |
| | | | 6 | 7 | 8 | | | | | | |
| 9 | 10 | 11 | 17 | 18 | 19 | 25 | 26 | 27 | 33 | 34 | 35 |
| 12 | White | 13 | 20 | Green | 21 | 28 | Blue | 29 | 36 | Yellow | 37 |
| 14 | 15 | 16 | 22 | 23 | 24 | 30 | 31 | 32 | 38 | 39 | 40 |
| | | | 41 | 42 | 43 | | | | | | |
| | | | 44 | Orange | 45 | | | | | | |
| | | | 46 | 47 | 48 | | | | | | |

| | | | | | | | | |
|-------|-------|------|--------|--------|--|--|--|--|
| | | | | Red | | | | |
| White | Green | Blue | Yellow | | | | | |
| | | | | Orange | | | | |

